THE UNIVERSITY OF MICHIGAN
INDUSTRY PROGRAM OF THE COLLEGE OF ENGINEERING

WATERHAMMER ANALYSIS WITH NONLINEAR FRICTIONAL RESISTANCE

V. L. Streeter

August, 1962

IP - 579

ACKNOWLEDGEMENT

Use of the facilities of the Computing Center of the University of Michigan is gratefully acknowledged.

TABLE OF CONTENTS

LIST OF FIGURES

INTRODUCTION

Closure of a valve in a pipe network may cause a rupture a short time later at a point remote from the valve. Owing to the reflections from junctions and dead ends, the maximum pressure at a point in the system may not be attained at passage of the first pressure rise due to closure, and numerical methods of analysis formerly used[1,2,3,4] are tedious to apply to complex systems, even with the neglect of fluid friction. The new methods presented here, worked out by Streeter and Lai,[5] include friction and are based on the theory of characteristics with use of a high-speed digital computer. Courant and Friedrichs[6] first applied these methods of computation to supersonic flow.

The equation of continuity and the equation of motion are first derived in this paper. The resulting nonlinear partial differential equations are then transformed into total differential equations which are valid along certain "characteristic" lines. These equations are expressed as finite difference equations, and interpolations are used to apply the method of specified time intervals. Boundary conditions for a reservoir, a valve, and pipe junctions are then discussed, and the methods applied to three complex pipe situations: a) a series pipe system with a dead-end branch; b) a parallel pipe system; and c) a pipe network with three simple circuits.

DEVELOPMENT OF EQUATIONS

The continuity equation and the equation of motion are developed for a segment of an elastic tube, assuming that deformations are small, as is the usual case with metal pipes.

Continuity Equation

The continuity equation is written for a short segment of pipe, Figure 1, of unstressed length Δx. When subjected to a change in pressure within the pipe, the walls are stressed both circumferentially and axially causing the segment to change in diameter and length. With μ the Poisson ratio,[7] σ_1 the axial stress and σ_2 the circumferential stress, the unit axial strain ξ_1 due to change in stresses $\Delta\sigma_1$ and $\Delta\sigma_2$ is

$$\xi_1 = \frac{\Delta\sigma_1 - \mu\,\Delta\sigma_2}{E}$$

and the unit circumferential strain ξ_2 is

$$\xi_2 = \frac{\Delta\sigma_2 - \mu\,\Delta\sigma_1}{E}$$

in which E is Young's modulus of elasticity. The increase in volume of the pipe element is then

$$A\xi_1\,\Delta x + \pi D\,\frac{D}{2}\,\xi_2\,\Delta x$$

neglecting second order terms. A is the cross-sectional area of the pipe and D is the inside diameter.

The extra volume that can be stored in the segment because of compressibility of the liquid is

$$\frac{A\,\Delta x\,\rho g\,\Delta H}{K}$$

in which ρ is the mass density of liquid, g the acceleration of gravity, K the bulk modulus of elasticity of liquid, and ΔH is the increase in head in time Δt. The net inflow into the element in time Δt is then the sum of the last two expressions

$$A \, \Delta x(\xi_1 + \frac{\xi_2}{2} + \frac{\rho g \, \Delta H}{K})$$

and is, from Figure 1b, equal to

$$-A \, V_x \, \Delta x \, \Delta t$$

in which V is the velocity at distance x along the pipe, and the subscript x denotes partial differentiation with respect to x.

After equating the last two expressions and after substituting for ξ_1 and ξ_2 their values,

$$\frac{1}{E} [\Delta\sigma_2(2 - \mu) + \Delta\sigma_1(1 - 2\mu)] + \frac{\rho g \, \Delta H}{K} + V_x \, \Delta t = 0 \qquad (1)$$

Three cases of pipe restraint[2] are considered (Figure 1a):

a) Pipe fixed at upstream end, free otherwise,

$$\Delta\sigma_1 = \frac{\rho g \, \Delta H \, D}{4t'} \qquad \Delta\sigma_2 = \frac{\rho g \, \Delta H \, D}{2t'}$$

in which t' is the pipe wall thickness. After making these substitutions into Equation (1) and by writing dH/dt in place of $\Delta H/\Delta t$,

$$\frac{\rho g}{K} \frac{dH}{dt} [1 + \frac{KD}{Et'} (\frac{5}{4} - \mu)] + V_x = 0 \qquad (2)$$

b) Pipe anchored throughout its length,

$$\Delta\sigma_1 = \mu \, \Delta\sigma_2 \qquad \Delta\sigma_2 = \frac{\rho g \, \Delta H \, D}{2t'}$$

and

$$\frac{\rho g}{K}\frac{dH}{dt}\left[1 + \frac{KD}{Et'}(1 - \mu^2)\right] + V_x = 0 \tag{3}$$

c) Expansion joints throughout,

$$\Delta\sigma_1 = 0 \qquad\qquad \Delta\sigma_2 = \frac{\rho g\ \Delta H\ D}{2t'}$$

and

$$\frac{\rho g}{K}\frac{dH}{dt}\left[1 + \frac{KD}{Et'}(1 - \frac{\mu}{2})\right] + V_x = 0 \tag{4}$$

By defining c_1 as the modifying expression involving the Poisson ratio (i.e., $c_1 = \frac{5}{4} - \mu$ for case a) the continuity equation may take the form

$$\frac{dH}{dt} + \frac{a^2}{g} V_x = 0 \tag{5}$$

in which

$$a^2 = \sqrt{\frac{K}{\rho(1 + \frac{c_1\ KD}{Et'})}} \tag{6}$$

a is the speed of the pressure pulse through the pipe.

Now, by expanding dH/dt

$$H_x \frac{dx}{dt} + H_t + \frac{a^2}{g} V_x = 0 \tag{7}$$

The first term may be shown to be small when compared with the second term,[2] and is dropped from the equation, leaving

$$H_t + \frac{a^2}{g} V_x = 0 \tag{8}$$

Equation of Motion

The equation of motion, Figure 2, for the x-direction states that the resultant x-component of force equals the product of the mass of the element by its acceleration,

$$pA - [pA + (pA)_x \Delta x] + pA_x \Delta x - \tau_o \pi D \Delta x = \rho A \Delta x \frac{dV}{dt}$$

τ_o is the fluid shear stress at the wall. After simplifying

$$p_x + \frac{4\tau_o}{D} + \rho \frac{dV}{dt} = 0$$

By replacing p_x by $\rho g H_x$, τ_o by $\rho f V^2/8$ to introduce the Darcy-Weisbach friction factor[8] f, and by expanding dV/dt

$$gH_x + \frac{fV^2}{2D} + VV_x + V_t = 0 \qquad (9)$$

The term VV_x may be shown[2] to be small compared with V_t and is dropped from the expression, yielding

$$gH_x + V_t + \frac{fV|V|}{2D} = 0 \qquad (10)$$

as the equation of motion. The absolute value sign is introduced into the frictional term so that it has the proper sign for reversal of flow direction in the pipe. If an exponential form of friction expression is desired in place of the quadratic term it may be substituted at any step in the derivation. If f is appropriate for the steady-state condition $V = V_o$, then an exponential form is

$$\frac{f}{2D} V|V|^{n-1}$$

with losses varying as V^n in the exponential equation.

Equations (8) and (10), labeled L_1 and L_2 respectively, are two hyperbolic partial differential equations, one nonlinear, that contains the continuity principle and Newton's second law of motion, plus elasticity and compressibility requirements. They are now solved by the method of characteristics.

Characteristic Equations

By combining Equations (8) and (10) linearly, using an unknown multiplier λ,

$$L = L_2 + \lambda L_1 = \lambda[\frac{g}{\lambda} H_x + H_t] + [\frac{\lambda a^2}{g} V_x + V_t] + \frac{f\, V|V|}{2D} = 0 \qquad (11)$$

By selecting any two arbitrary (non-equal) values of λ, L provides two new independent equations in V and H. However, by taking two particular values for λ, L may be transformed into a pair of total differential equations. In the first set of brackets in Equation (11), by making the restriction $g/\lambda = dx/dt$ the quantity in brackets becomes the total derivative dH/dt. Similarly, by making the restriction $dx/dt = \lambda a^2/g$, the quantity in the second set of brackets becomes dV/dt.

The two expressions for dx/dt must be identical, hence,

$$\frac{g}{\lambda} = \frac{\lambda a^2}{g}$$

After solving for λ, the unknown multiplier is

$$\lambda = \pm \frac{g}{a} \qquad (12)$$

Hence the restriction imposed on the equations is

$$\frac{dx}{dt} = \pm\, a \qquad (13)$$

The Equations (13) together with Equations (11) with λ given by Equation (12) are



Figure 1. Segment of Pipe for Application of Continuity Equation.



Figure 2. Segment of Pipe Showing Forces Exerted on Fluid in the x-Direction.



Figure 3. Characteristic Lines on the x - t Plane.

$$\left.\begin{aligned} \frac{g}{a}\frac{dH}{dt} + \frac{dV}{dt} + \frac{fV|V|}{2D} = 0 \qquad (14) \\ \frac{dx}{dt} = a \qquad (15) \end{aligned}\right\} C_+$$

$$\left.\begin{aligned} -\frac{g}{a}\frac{dH}{dt} + \frac{dV}{dt} + \frac{fV|V|}{2D} = 0 \qquad (16) \\ \frac{dx}{dt} = -a \qquad (17) \end{aligned}\right\} C_-$$

These four equations now replace Equations (8) and (10). Equation (14) is subject to the restriction of Equation (15). Since a is generally constant for a given pipe, Equation (15), $dx/dt = a$, plots as a straight line on an x - t plane; and similarly Equation (16) is limited to lines $dx/dt = -a$ on the x - t plane. These lines on the x - t plane are the "characteristic" lines along which Equations (14) and (16) are valid.

Equations (14) and (16) may be written in terms of finite differences, Figure 3, in which P is the unknown point and R and S are respectively points on Equations (15) and (17) that are known.

$$\frac{g}{a}(H_P - H_R) + V_P - V_R + \left(\frac{fV|V|}{2D}\right)_R (t_P - t_R) = 0 \qquad (18)$$

$$x_P - x_R - a(t_P - t_R) = 0 \qquad (18a)$$

$$-\frac{g}{a}(H_P - H_S) + V_P - V_S + \left(\frac{fV|V|}{2D}\right)_S (t_P - t_S) = 0 \qquad (19)$$

$$x_P - x_S + a(t_P - t_S) = 0 \qquad (19a)$$

If the pipe thickness varies along the pipe, a becomes a function of x. With conditions known at R and S (i.e., with x_R, t_R, V_R, H_R, x_S, t_S, V_S, and H_S known) the four equations permit solution for x_P, t_P, V_P, and H_P. In this manner solutions may be built up from known initial conditions and from end conditions.

Method of Specified Time Intervals

The procedure outlined for solution by the method of characteristics is not very suitable for systematic machine solution of transient flow problems. By means of interpolation,[9] one may find values of V_R, V_S, H_R, and H_S for points on the characteristic lines that go through specified points on the x - t plane, Figure 4. Consider that values of V and H are known for the evenly spaced points along the pipe, Δx apart, at time $t = t_C$. From the values of V and H at A, C, and B, values of V_R, H_R, V_S and H_S are computed by linear interpolation, as follows:

$$\frac{x_C - x_R}{x_C - x_A} = \frac{V_C - V_R}{V_C - V_A}$$

from Figure 4. By use of Equation (18a), since $x_P = x_C$

$$x_C - x_R = x_P - x_R = a(t_P - t_R) = a\ \Delta t$$

Now, solving for V_R

$$V_R = V_C + \Theta\ a(V_A - V_C) \tag{20}$$

in which Θ is $\Delta t/\Delta x$, the ratio of preselected time increment to distance increment. Similarly

$$H_R = H_C + \Theta\ a(H_A - H_C) \tag{21}$$

$$V_S = V_C + \Theta\ a\ (V_B - V_C) \tag{22}$$

$$H_S = H_C + \Theta\ a(H_B - H_C) \tag{23}$$

For convergence of the solution Δt should be less than $\Delta x/a$.

By use of the interpolated values, only two equations are needed, Equations (18) and (19) to solve for the two unknowns, H_P and V_P.

The friction terms in Equations (18) and (19) may be evaluated for conditions at C without appreciable error, if increments Δx and Δt are kept sufficiently small. After solving Equations (18) and (19) for H_P and V_P

$$H_P = \frac{H_R + H_S}{2} + \frac{a}{2g} (V_R - V_S) \tag{24}$$

$$V_P = \frac{V_R + V_S}{2} + \frac{g}{2a} (H_R - H_S) - \frac{f_c \, V_c |V_c|}{2D} \Delta t \tag{25}$$

These equations, together with Equations (20) to (23), permit computation of velocity and head at interior points in the pipe. Special relations for computation of end points are needed, known as boundary conditions, and are discussed in the following section.

Boundary Conditions

At the end of a pipe, only one of the two Equations (18) and (19) is applicable. At the upstream end ($x = 0$), the interpolation Equations (22) and (23) are appropriate, and with Equation (19), one equation is available in the two unknowns H_P and V_P. An auxiliary equation is needed that specifies V_P, H_P or some relation between them, so that two equations in two unknowns are available. The simplest case is probably that of a reservoir, where H_P is constant, or a function of time.

At the downstream end of a pipe Equations (18), (20), and (21) are available and an auxiliary relation is needed to solve for V_P and H_P. The relationships between V_P and H_P for a valve are worked out to illustrate the procedures for determining boundary conditions.

The valve is treated as an orifice. With the steady-state head loss H_o across the valve and the steady-state velocity V_o in the pipe, the orifice equation becomes

$$V_o A = (C_d\ A_G)_o \sqrt{2g\ H_o} \tag{26}$$

with A the pipe cross-sectional area, C_d the valve coefficient and A_G the valve opening. In general

$$V_P A = C_d\ A_G \sqrt{2g\ H_P} \tag{27}$$

After dividing Equation (27) by Equation (26) and substituting $\tau = C_d\ A_G/(C_d\ A_G)_o$, the dimensionless valve opening

$$\frac{V_P}{V_o} = \tau \sqrt{\frac{H_P}{H_o}} \tag{28}$$

In this dimensionless equation τ varies from 1 to 0 for closure, in a manner depending upon the rate of closure. τ is usually expressed as a function of time, as shown in Figure 5. Equation (28) provides the second relation between V_P and H_P. By eliminating H_P in Equations (18) and (28)

$$V_P = \frac{a\ \tau^2\ V_o^2}{2g\ H_o} \left[\sqrt{1 + \frac{4g\ H_o}{a\tau^2\ V_o^2} \left(\frac{g\ H_R}{a} + V_R - \frac{f_c\ V_c |V_c|\ \Delta t}{2D}\right)} - 1 \right] \tag{29}$$

In the solution of the quadratic in V_P the positive sign was taken before the radical so that $V_P = V_o$ for $\tau = 1$ in the steady-state case. After V_P is determined from Equation (29) H_P is found from Equation (28). For the special case $\tau = 0$, $V_P = 0$, and H_P is found from Equation (18).

In general, boundary conditions are easy to formulate. Special cases are included in the applications to complex piping situations.



Figure 4. Preselected Points on the xt-plane for which Calculations of H and V are to be made.



Figure 5. Valve Closing Relationship.

APPLICATIONS

Three applications are presented that show the general adaptability of equations and the boundary conditions. The Poisson's ratio effects have not been specifically taken into account, since its inclusion affects only the speed of pressure wave, and the pipe thicknesses have been arbitrarily assumed for the examples. First a series pipe system with a deadend branch is taken up, followed by a 4 pipe system with two of the pipes in parallel, and concluding with a network having 9 pipes and 3 simple circuits.

Series Pipe System with Dead-End Branch

In Figure 6 a series pipe system with a dead-end branch is taken for the first example. The valve closure is

$$\tau = 1 - \sqrt{\frac{t}{t_c}}$$

in which $t_c = 0.518$ sec. is the time of closure of the valve. It is the round-trip travel time of a pressure pulse in pipe 1. The computer program, written in MAD (Michigan Algorithm Decoder) language, together with a sample of the calculations, is shown in Figure 7. A plot of head versus time for the dead end of pipe 2, the junction of pipes 1 and 3, and the valve are shown in Figure 8. Constant friction factors have been assumed [F(1) ... F(3)], which results in V^2 losses. The steady-state solution is first obtained with $\tau = 1$, and the head and velocity at 11 equally-spaced sections along each pipe are computed, stored and printed for time t = 0. The time is then incremented (DELT) by an amount such that the interpolation points R and S lie within A and C, Figure 4, for all three pipes. The interior points are next calculated,



Figure 6. Series Pipe System with Dead-End Branch. Steel Pipe.

```
          RWATERHAMMER IN SERIES PIPE WITH DEAD END BRANCH
           INTEGER I,J,N,U,W
           DIMENSION L(3),D(3),F(3),A(3),THI(3),TH(3),HF(3),Q(3),LA(3),H
          2RR(3),V7(3),HJ(300),HV(300),TT(300),V(99,DIM),VP(99,DIM),H(99
          3,DIM),HP(99,DIM),DIM(2)
           VECTOR VALUES DIM=2,26,11
HERE       READ DATA
           PRINT COMMENT$1                                  GIVEN DATA$
           PRINT RESULTS Q(1),F(1),F(2),F(3),L(1),L(2),L(3),D(1),D(2),D(
          23),THI(1),THI(2),THI(3),HO,RHO,E,K,N,W,G,CON
          R
          RSTEADY STATE RELATIONSHIPS
           LAMIN=10.
           THROUGH Z1,FOR I=1,1,I.G.3
           V7(I)=Q(I)/(.7854*D(I)*D(I))
           HF(I)=F(I)*L(I)*V7(I)*V7(I)/(2.*G*D(I))
           A(I)=SQRT.(K/(RHO*(1.+K*D(I)/(E*THI(I)))))
           LA(I)=L(I)/A(I)
           WHENEVER LA(I).L.LAMIN
           LAMIN=LA(I)
           J=I
Z1         END OF CONDITIONAL
           HF=HF(1)+HF(3)
           HRR=HO+HF
           TC=2.*LA(J)*CON
           DELT=LA(J)/N
           PRINT RESULTS HRR,DELT,HF(I),A(1),A(2),A(3),LA(1),LA(2),LA(3)
          2,TC
           HRR(3)=HRR
           HRR(1)=HRR-HF(3)
           HRR(2)=HO
           HJ=HRR(1)
           HV=HO
           EXECUTE ZERO.(T,U,TT)
           TAU=1.
           THROUGH Z2,FOR J=1,1,J.G.3
           TH(J)=DELT*N/L(J)
           THROUGH Z2,FOR I=0,1,I.G.N
           V(J,I)=V7(J)
Z2         H(J,I)=HRR(J)-HF(J)*I/N
           HP(3,0)=HRR
           PRINT COMMENT$1   VELOCITIES AND HEADS AT TENTH POINTS FOR EQ
          2UAL TIME INCREMENTS$
           PRINT COMMENT$OPIPE TIME   TAU   X/L=      .0      .1      .
          22      .3      .4      .5      .6      .7      .8      .9
          3 1.$
Z3         THROUGH Z5,FOR J=1,1,J.G.3
Z5         PRINT FORMAT Z4,J,T,TAU,H(J,0)...H(J,N),J,V(J,0)...V(J,N)
           VECTOR VALUES Z4=$1H0,I3,2F6.3,S6,4H H= ,11F8.3/1H ,I3,S18,4H
          2 V= ,11F8.3*$
Z6         T=T+DELT
           U=U+1
           WHENEVER U.E.W,TRANSFER TO PLOT
          R
          RCOMPUTATION OF INTERIOR POINTS
           THROUGH Z9,FOR J=1,1,J.G.3
```

```
          THROUGH Z9,FOR I=1,1,I.E.N
          VR=V(J,I)+TH(J)*A(J)*(V(J,I-1)-V(J,I))
          HR=H(J,I)+TH(J)*A(J)*(H(J,I-1)-H(J,I))
          HS=H(J,I)+TH(J)*A(J)*(H(J,I+1)-H(J,I))
          VS=V(J,I)+TH(J)*A(J)*(V(J,I+1)-V(J,I))
          VP(J,I)=(VR+VS)/2.+16.1*(HR-HS)/A(J)-F(J)*V(J,I)*.ABS.(V(J,I)
         2)*DELT/(2.*D(J))
Z9        HP(J,I)=A(J)*(VR-VS)/64.4+(HR+HS)/2.
         R
         RBOUNDARY CONDITION AT RESERVOIR
          VS=V(3,0)+TH(3)*A(3)*(V(3,1)-V(3,0))
          HS=H(3,0)+TH(3)*A(3)*(H(3,1)-H(3,0))
          VP(3,0)=VS+G*(H(3,0)-HS)/A(3)-F(3)*V(3,0)*.ABS.V(3,0)*DELT/(2
         2.*D(3))
         R
         RJUNCTION BOUNDARY CONDITION
          VS=V(1,0)+TH(1)*A(1)*(V(1,1)-V(1,0))
          HS=H(1,0)+TH(1)*A(1)*(H(1,1)-H(1,0))
          VR=V(3,N)+TH(3)*A(3)*(V(3,N-1)-V(3,N))
          HR=H(3,N)+TH(3)*A(3)*(H(3,N-1)-H(3,N))
          C1=VS-G*HS/A(1)-F(1)*V(1,0)*.ABS.V(1,0)*DELT/(2.*D(1))
          C2=VR+G*HR/A(3)-F(3)*V(3,N)*.ABS.V(3,N)*DELT/(2.*D(3))
          HP(3,N)=(C2-C1*D(1)*D(1)/(D(3)*D(3)))/(G/A(3)+G*D(1)*D(1)/(D(
         23)*D(3)*A(1)))
          HP(1,0)=HP(3,N)
          VP(3,N)=C2-G*HP(3,N)/A(3)
          VP(1,0)=C1+G*HP(3,N)/A(1)
         R
         RDEAD END BOUNDARY CONDITION
          VP(2,N)=0.
          VR=V(2,N)+TH(2)*A(2)*(V(2,N-1)-V(2,N))
          HR=H(2,N)+TH(2)*A(2)*(H(2,N-1)-H(2,N))
          HP(2,N)=HR+A(2)*VR/G
         R
         RBOUNDARY CONDITION AT VALVE
          WHENEVER T.LE.TC,TAU=1.-SQRT.(T/TC)
          WHENEVER T.G.TC,TAU=0.
          VR=V(1,N)+TH(1)*A(1)*(V(1,N-1)-V(1,N))
          HR=H(1,N)+TH(1)*A(1)*(H(1,N-1)-H(1,N))
          VS=V(2,0)+TH(2)*A(2)*(V(2,1)-V(2,0))
          HS=H(2,0)+TH(2)*A(2)*(H(2,1)-H(2,0))
          C1=VS-G*HS/A(2)-F(2)*V(2,0)*.ABS.V(2,0)*DELT/(2.*D(2))
          C2=G/A(2)
          C3=VR+G*HR/A(1)-F(1)*V(1,N)*.ABS.V(1,N)*DELT/(2.*D(1))
          C4=-G/A(1)
          C6=(C3-C1*D(2)*D(2)/(D(1)*D(1)))/(C4-C2*D(2)*D(2)/(D(1)*D(1))
         2)
          WHENEVER TAU.E.0.,HP(1,N)=-C6
          WHENEVER TAU.G.0.
          C5=TAU*V7(1)/((C4-C2*D(2)*D(2)/(D(1)*D(1)))*SQRT.(H0))
          C7=SQRT.(C5*C5/4.-C6)
          HP(1,N)=(C5/2.+C7)*(C5/2.+C7)
          END OF CONDITIONAL
          HP(2,0)=HP(1,N)
          VP(1,N)=C3+C4*HP(1,N)
          VP(2,0)=C1+C2*HP(1,N)
         R
         RSTORAGE OF PLOTTING INFORMATION
```

```
              TT(U)=TT(U-1)+DELT
              HJ(U)=HP(3,N)
              HV(U)=HP(1,N)
              THROUGH Z10,FOR J=1,1,J.G.3
              THROUGH Z10,FOR I=0,1,I.G.N
              H(J,I)=HP(J,I)
Z10           V(J,I)=VP(J,I)
              WHENEVER U/5*5.E.U,TRANSFER TO Z3
              TRANSFER TO Z6
PLOT          PRINT RESULTS HJ...HJ(W-1),HV...HV(W-1)
              PRINT COMMENTS1                 HEAD AT VALVE  VS  TIME$
              EXECUTE SETPLT.(1,TT,HV,W,S*S,24,ORD)
              VECTOR VALUES ORD=$                    HEAD$
              PRINT COMMENTS0                   TIME IN SECONDS$
              PRINT COMMENTS1           HEAD AT JUNCTION BETWEEN PIPES 1 AND
             2 3  VS TIME$
              EXECUTE SETPLT.(1,TT,HJ,W,S*S,24,ORD)
              PRINT COMMENTS0                   TIME IN SECONDS$
              TRANSFER TO HERE
              END OF PROGRAM
$DATA
L(1)=1000.,2000.,1500.,D(1)=3.,2.,2.5,F(1)=.025,.03,.03,THI(1)=.06,.05,
 .06,H0=600.,RHO=1.935,E=4.32E9,K=4.32E7,N=10,W=300,G=32.2,CON=1.,
Q(1)=20.,0.,20.*
```

GIVEN DATA

Q(1) = 2C.000000,	F(1) = .025000,	F(2) = .030000,	F(3) = .030000
L(1) = 100C.000000,	L(2) = 2000.000000,	L(3) = 1500.000000,	D(1) = 3.000000
D(2) = 2.000000,	D(3) = 2.500000,	THI(1) = .060000,	THI(2) = .050000
THI(3) = .060000,	HO = 600.000000,	RHO = 1.935000,	E = 4.320000E 09
K = 4.32C000E 07,	N = 10,	W = 300,	G = 32.200000
CON = 1.00000C			
HRR(0) = 605.675774,	DELT = .025921,	HF(1) = 1.035920,	A(1) = 3857.942596
A(2) = 3993.349976,	A(3) = 3969.790161,	LA(1) = .259206,	LA(2) = .500833
LA(3) = .377854,	TC = .518411		

VELOCITIES AND HEADS AT TENTH POINTS FOR EQUAL TIME INCREMENTS

PIPE	TIME	TAU	X/L=	.0	.1	.2	.3	.4	.5	.6	.7	.8	.9	1.
1	.000	1.000	H=	601.036	600.932	600.829	600.725	600.622	600.518	600.414	600.311	600.207	600.104	600.000
1			V=	2.829	2.829	2.829	2.829	2.829	2.829	2.829	2.829	2.829	2.829	2.829
2	.000	1.000	H=	600.000	600.000	600.000	600.000	600.000	600.000	600.000	600.000	600.000	600.000	600.000
2			V=	.000	.000	.000	.000	.000	.000	.000	.000	.000	.000	.000
3	.000	1.000	H=	605.676	605.212	604.748	604.284	603.820	603.356	602.892	602.428	601.964	601.500	601.036
3			V=	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074
1	.130	.500	H=	601.036	600.932	600.829	600.725	600.621	600.518	646.505	666.565	682.413	696.098	708.384
1			V=	2.829	2.829	2.829	2.829	2.829	2.829	2.445	2.277	2.144	2.029	1.926
2	.130	.500	H=	708.384	680.888	647.781	618.438	603.307	600.000	600.000	600.000	600.000	600.000	600.000
2			V=	.874	.652	.385	.149	.027	.000	.000	.000	.000	.000	.000
3	.130	.500	H=	605.676	605.212	604.748	604.284	603.820	603.356	602.892	602.428	601.964	601.500	601.036
3			V=	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074
1	.259	.293	H=	601.036	647.022	667.017	682.840	696.505	708.776	720.082	730.637	740.622	750.121	759.237
1			V=	2.829	2.445	2.277	2.144	2.030	1.927	1.832	1.743	1.659	1.579	1.503
2	.259	.293	H=	759.237	740.537	719.338	694.583	666.731	639.447	618.216	606.032	601.257	600.123	600.000
2			V=	1.283	1.133	.962	.763	.538	.318	.147	.049	.010	.001	.000
3	.259	.293	H=	605.676	605.212	604.748	604.284	603.820	603.356	602.892	602.428	601.964	601.500	601.036
3			V=	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074	4.074
1	.389	.134	H=	730.216	739.090	746.955	753.830	759.415	759.577	768.336	776.820	785.038	793.046	800.841
1			V=	2.103	1.988	1.877	1.768	1.655	1.504	1.430	1.359	1.290	1.223	1.157
2	.389	.134	H=	800.841	785.188	768.527	750.440	730.295	707.540	682.453	656.968	634.625	619.339	613.918
2			V=	1.618	1.492	1.358	1.212	1.050	.867	.665	.458	.271	.121	.000
3	.389	.134	H=	605.676	605.212	604.748	604.284	603.820	603.356	615.080	644.641	679.030	707.559	730.216
3			V=	4.074	4.074	4.074	4.074	4.074	4.074	3.975	3.732	3.450	3.215	3.029
1	.518	-.000	H=	790.899	797.850	804.444	810.689	816.606	822.177	827.388	832.164	836.385	839.659	837.773
1			V=	1.764	1.675	1.588	1.503	1.418	1.334	1.250	1.165	1.078	.985	.851
2	.518	-.000	H=	837.773	823.713	809.114	793.844	777.786	760.932	743.634	726.886	712.448	702.512	698.953
2			V=	1.914	1.801	1.684	1.561	1.429	1.284	1.114	.907	.649	.341	.000
3	.518	-.000	H=	605.676	607.063	615.390	634.504	661.826	690.746	716.613	738.587	757.677	774.910	790.899
3			V=	4.074	4.059	3.988	3.829	3.604	3.366	3.154	2.972	2.815	2.672	2.540
1	.648	.000	H=	840.615	846.677	852.534	858.209	863.697	869.013	870.768	870.441	869.701	868.694	867.457
1			V=	1.487	1.410	1.334	1.259	1.185	1.111	1.066	1.037	1.009	.983	.956
2	.648	.000	H=	867.457	859.696	848.976	836.197	824.739	817.013	813.305	813.113	815.225	817.706	818.774
2			V=	2.150	2.084	1.987	1.853	1.695	1.510	1.286	1.015	.703	.359	.000
3	.648	.000	H=	605.676	653.004	693.180	723.680	746.667	765.551	782.437	798.125	812.932	827.054	840.615
3			V=	3.352	3.315	3.215	3.071	2.914	2.763	2.623	2.493	2.370	2.253	2.141

Figure 7. Computer Program and Sample Calculations for Series Pipe with Dead-End Branch.



Figure 8. Head Versus Time at Junction, Valve and Dead End, for Series Pipe System of Figure 6. $\tau = 1 - \sqrt{t/t_c}$, $t_c = 0.518$ sec.

then the various boundary conditions. At the reservoir the head [H(3,0)] is held constant, and the velocity [VP(3,0)] is calculated from Equation (19). At the junction between pipes minor losses are neglected, and the conditions to be satisfied are continuity and equality of head (this neglects change in velocity head, as is customarily done). At the dead-end the velocity is zero [VP(2,N) = 0], and the head is computed from Equation (18). The friction term drops out of statement 067 (Figure 7) since $V_C = 0$.

At the valve, the boundary conditions are continuity and equality of head. For continuity

$$V_1 A_1 - V_2 A_2 = C_D A_G \sqrt{2g H}$$

and for steady flow

$$V_o A_1 = (C_D A_G)_o \sqrt{2g H_o}$$

After dividing the first equation by the second equation

$$\frac{V_1}{V_o} - \frac{V_2}{V_o}\frac{A_2}{A_1} = \tau\sqrt{\frac{H}{H_o}}$$

By substituting for V_1 and V_2 from Equations (18) and (19), a quadratic in $\sqrt{H}$ is obtained. The proper root of the radical term in the quadratic solution must be selected so that when $\sqrt{H}$ is squared it equals H_o in the limit for steady flow.

This completes the calculation of all points for this time. These values are stored temporarily, and after 5 time increments, the results are printed. Figure 8 illustrates the complex variation of head with time due to the reflections from the various boundaries.

Parallel Pipe System

A parallel pipe system, Figure 9, is the second example of waterhammer in a complex circuit. Boundary conditions must be worked out for the reservoir, the upstream and downstream junctions, and the valve. At the reservoir the head remains constant, so Equation (19) provides the value of V_P after VS and HS are computed. The junction boundary conditions are continuity and a common head for the 3 pipes. For the downstream junction Equation (18) is applied twice, to pipes 2 and 3, and Equation (19) to pipe 4. With continuity this yields 4 equations in 4 unknowns, V_{P_2} , V_{P_3} , V_{P_4} , and H_P. A similar procedure applies to the upstream junction. For the value Equation (30) may be applied when $\tau > 0$. For $\tau = 0$, $V_P = 0$, and H_P is given by Equation (18).

Owing to space limitations, the program is not presented, but head versus time for the two junctions and the valve are presented in Figure 10, with initial head at the valve $H_O = 500$ ft. and for instantaneous valve closure.

Pipe Network

As the final example of flow through a complex system, a network, Figure 11, is selected having 9 pipes and 3 simple circuits. A valve at the downstream end is closed uniformly in twice the round-trip travel time of the pressure wave through pipe 9. The steady-state flow through the system is first determined by the Hardy Cross method[10] of distributing flow in a network, and the results are shown in Figure 11.



Figure 9. Parallel Pipe System. Steel Pipe; Q_1 = 20 cfs., Q_2 = 8.39 cfs.



Figure 10. Plot of Head Versus Time for Sudden Valve Closure for the Pipe System of Figure 9.



Figure 11. Pipe Network for Waterhammer Calculation. Steel Pipe.

In writing the watehammer program double subscripting is used, i.e., V(3,4) is the velocity in pipe 3 at the fourth section. The velocities and heads were computed for 9 equally-spaced sections in each pipe (i.e., 250 ft. apart in pipe 1, and 375 ft. apart in pipe 2), and the time increment $\Delta t = 0.04695$ sec. was selected on the basis of pipe 8 calculations so that Δt does not exceed $\Delta x/a$ in any pipe.

The boundary conditions at the reservoir, the valve, and the 5 interior junctions were satisfied for each time increment. The head was held constant at the reservoir, and at each junction continuity was satisfied and the head made the same for end sections of each pipe entering or leaving the junction. The program is not included to conserve space, but plots of head versus time at the 5 interior junctions and the valve are shown in Figures 12 and 13.



Figure 12. Head Versus Time Plots for Junctions J1, J2, and J3 for Network of Figure 11 for Uniform Valve Closure in 2 Seconds.



Figure 13. Head Versus Time Plots for Junctions J4, J5, and Valve for Network of Figure 11 for Uniform Valve Closure in 2 Seconds.

SUMMARY

Equations are derived for the numerical solution of transient liquid flow in a conduit with nonlinear friction by use of the theory of characteristics and the method of specified time intervals. Boundary conditions for various cases have been discussed to show their ease of formulation. The methods are applied to 3 complex pipe systems: series, parallel, and network, using an IBM 709 high-speed digital computer to generate solutions which are presented as plots of head against time.

REFERENCES

1. Allievi, L., "Theory of Waterhammer," translated by E. E. Halmos, printed by Richardo Garoni, Rome, Italy, 1925.

2. Parmakian, J., "Waterhammer Analysis," Prentice-Hall, Inc., New York, 1955.

3. Angus, R. W., "Water-Hammer Pressures in Compound and Branched Pipes," Trans. ASCE, Vol. 104, pp. 340-401 (1959).

4. Rich, G. R., "Hydraulic Transients," McGraw-Hill Book Co., Inc., New York, 1951.

5. Streeter, V. L., and Chintu, Lai, "Water-Hammer Analysis Including Fluid Friction," J. Hyd. Div. ASCE, Vol. 88, No. HY3, May, 1962, Part I.

6. Courant, R., and Friedrichs, K. O., "Supersonic Flow and Shock Waves," Interscience Publishers, New York, 1948.

7. Timoshenko, S., Strength of Materials, 2nd Ed., Part 2, D. Van Nostrand, Inc., New York, 1941.

8. Streeter, V. L., "Fluid Mechanics," 3rd Ed., McGraw-Hill Book Company, Inc., New York, pp. 211, 214-222, (1962).

9. Lister, M., The Numerical Solution of Hyperbolic Partial Differential Equations by the Method of Characteristics in "Mathematical Methods for Digital Computers," edited by Anthony Ralston and Herbert S. Wilf, John Wiley and Sons, Inc., 1960.

10. Cross, Hardy, "Analysis of Flow in Networks of Conduits or Conductors, University of Illinois, Bull. 286, November 1946.